SCORING INSTRUCTIONS (from the psych manual)
To score the BFI, you’ll first need to reverse-score all negatively-keyed items:

Extraversion: 6, 21, 31

Agreeableness: 2, 12, 27, 37

Conscientiousness: 8, 18, 23, 43

Neuroticism: 9, 24, 34

Openness: 35, 41

To recode these items, you should subtract your score for all reverse-scored items from 6. For example, if you gave yourself a 5, compute 6 minus 5 and your recoded score is 1. That is, a score of 1 becomes 5, 2 becomes 4, 3 remains 3, 4 becomes 2, and 5 becomes 1.

Next, you will create scale scores by averaging the following items for each B5 domain (where R indicates using the reverse-scored item).

Extraversion: 1, 6R 11, 16, 21R, 26, 31R, 36

Agreeableness: 2R, 7, 12R, 17, 22, 27R, 32, 37R, 42

Conscientiousness: 3, 8R, 13, 18R, 23R, 28, 33, 38, 43R

Neuroticism: 4, 9R, 14, 19, 24R, 29, 34R, 39

Openness: 5, 10, 15, 20, 25, 30, 35R, 40, 41R, 44

SPSS SYNTAX

*** REVERSED ITEMS

RECODE

  bfi2 bfi6 bfi8 bfi9 bfi12 bfi18 bfi21 bfi23 bfi24 bfi27 bfi31 bfi34 bfi35

  bfi37 bfi41 bfi43

  (1=5)  (2=4)  (3=3)  (4=2)  (5=1)  INTO  bfi2r bfi6r bfi8r bfi9r bfi12r bfi18r bfi21r bfi23r bfi24r 

  bfi27r bfi31r bfi34r bfi35r bfi37r bfi41r bfi43r.

EXECUTE .

*** SCALE SCORES

COMPUTE bfie = mean(bfi1,bfi6r,bfi11,bfi16,bfi21r,bfi26,bfi31r,bfi36) .

VARIABLE LABELS bfie 'BFI Extraversion scale score.

EXECUTE .

COMPUTE bfia = mean(bfi2r,bfi7,bfi12r,bfi17,bfi22,bfi27r,bfi32,bfi37r,bfi42) .

VARIABLE LABELS bfia 'BFI Agreeableness scale score' .

EXECUTE .

COMPUTE bfic = mean(bfi3,bfi8r,bfi13,bfi18r,bfi23r,bfi28,bfi33,bfi38,bfi43r) .

VARIABLE LABELS bfic 'BFI Conscientiousness scale score' .

EXECUTE .

COMPUTE bfin = mean(bfi4,bfi9r,bfi14,bfi19,bfi24r,bfi29,bfi34r,bfi39) .

VARIABLE LABELS bfin 'BFI Neuroticism scale score' .

EXECUTE .

COMPUTE bfio = mean(bfi5,bfi10,bfi15,bfi20,bfi25,bfi30,bfi35r,bfi40,bfi41r,bfi44) .

VARIABLE LABELS bfio 'BFI Openness scale score' .

EXECUTE .

